
El modulo  719 – Reloj en tiempo real- utiliza el circuito 
integrado DS1302, que es un reloj/calendario con la 
capacidad de almacenar los datos correspondientes a las 
horas, minutos, segundos, día de la semana, día del mes, el 
mes y el año. Este dispositivo contiene lo necesario en su 
interior para actualizar automáticamente los datos en sus 
registros como reloj. 
El DS1302 utiliza un oscilador, basado en un cristal de 
cuarzo externo de 32.768KHz.,  para marcar el paso cada 
segundo y actualizar los datos en sus registros internos. 
Una interfase de comunicación tipo serie síncrona se utiliza 
para comunicar al DS1302 con un microcontrolador, ya sea 
para leer o escribir los datos que almacena o bien  
configurar la operación del circuito integrado. 
El DS1302 también incluye una memoria interna de 31 
bytes de uso general. 

TERMINALES DEL DS1302 
 
Pin Nombre Función 
1 Vcc2 Fuente de alimentación primaria. Alimenta al DS1302 mientras 

Vcc2 > Vcc1 + 0.2.  Cuando Vcc2<Vcc1, entonces Vcc1 alimenta 
al integrado 

2 X1 
3 X2 

Terminales para el cristal de cuarzo de 32.768KHz 

4 GND Tierra 
5 CE ó RST Terminal de entrada. Mientras sea “1” lógico se habilita la 

comunicación serie del DS1302 para lectura o escritura. Cuando 
CE=0 no se deshabilita cualquier comunicación.  

6 IO Terminal de entrada o salida de datos. Se utiliza para la 
comunicación serie. 

7 SCLK Señal de entrada de reloj para la sincronía de las comunicaciones 
8 Vcc1 Fuente de alimentación secundaria de baja potencia. Entra 

automáticamente cuando Vcc2<Vcc1. Se usa solo para mantener 
el reloj activo. También se utiliza para recargar una batería, que 
en este caso es Vcc2. 

 
    Para realizar una operación de lectura o escritura del 
DS1302, se requieren utilizar comandos cuya estructura 
indica la naturaleza de la instrucción, si es de lectura o 
escritura y sobre que registro va dirigida la información o 
de cual se desea leer un dato. 
El formato general de los comandos se muestra a 
continuación 

.  
Tome en cuenta lo siguiente: 

a) El bit 7 siempre es 1. 
b) El bit 6 indica si la operación es sobre la RAM o sobre 

los registros del reloj/calendario. Para operaciones 
sobre la RAM de uso general el bit 6 debe ser “1” y 
para operaciones sobre el reloj el bit 6 debe ser “0” 

c) Los bits 4,3,2,1 (A4, A3, A2,A1) indican la dirección del 
registro sobre el cual va dirigida la instrucción (lectura 
o escritura) 

d) El bit 0 indica si la operación es de lectura o escritura, 
en el caso particular debe ser 1 para una operación de 
lectura y  0 para un operación de escritura 

 
 
RESUMEN DE LOS COMANDOS 
 

  Tabla1 

Los números hexadecimales de las primeras dos columnas 
son los comandos que operan sobre el registro de memoria 
ya sea para lectura o escritura.  
 
● El bit 7 del registro de los segundos se utiliza para 
detener o iniciar la operación del reloj. Si el bit 7  es cero el 
reloj opera normalmente. Si el bit 7 es 1, la operación del 
reloj se detiene. 
● El  bit 7 (WP, WriteProtect) del registro de control se 
utiliza para habilitar o deshabilitar la escritura hacia el 
DS1302. Si el bit 7 es 1 la escritura se deshabilita. Si el bit 7 
es cero, la escritura se habilita.  
 
MODO BURST 
El modo Burst permite acceder de manera continua, bit por 
bit, a todos los registros del reloj/calendario o de la 
memoria RAM empezando por el bit 0 de la localidad 0. 
 
MODO DE INGRESO DE LOS COMANDOS  
La interfase de comunicaciones utiliza 3 terminales que son: 
RST, IO y SCLK. Para cualquier comunicación, la terminal 
RST debe ser “1”, de lo contrario no se habilita la 
comunicación. 
 
El envío o recepción de los comandos/direcciones hacia o 
desde el DS1302 se realiza bit a bit por la terminal IO con 
cada flanco de subida de la señal SCLK. La terminal IO es 
bidireccional ya que por ella puede fluir la información 
desde o hacia el DS1302. 

Para cualquier transferencia de información la terminal RST 
debe ponerse a “1”. Cuando se finalice la transferencia de 
información, se recomienda devolver a la terminal RST a 
“0”, esto es para que el próximo ingreso de datos sea un 
comando, el cual definirá una nueva instrucción. 
 
OPERACIÓN DE LECTURA 
Para leer los registros del DS1302 se debe enviar un 
comando valido de lectura  de acuerdo a los códigos 
mostrados en la Tabla 1. El comando se envía  bit por bit 
sobre la terminal IO empezando con el bit 0 con los flancos 
de subida en SCLK. La terminal IO se vuelve salida tras los 
primeros 8 flancos de bajada en SCLK y la información del 
byte a leer estará disponible en los siguientes 8 flancos de 
bajada en SCLK de la señal de reloj, empezando desde el 
bit 0. Cada flanco manifiesta un bit de información en IO.  
 
OPERACIÓN DE ESCRITURA. 
Para escribir los registros del DS1302 se debe enviar un 
comando valido de escritura  de acuerdo a los códigos 
mostrados en la Tabla 1. El comando se envía  bit por bit 
sobre la terminal IO empezando con el bit 0 con los flancos 
de subida en SCLK. La terminal IO permanece como 
entrada tras los primeros 8 flancos de subida y la escritura 
de localidad de memoria  se realiza  en los siguientes 8 
flancos de subida de la señal de reloj, empezando desde el 
bit 0 del dato que se desea enviar. Cada flanco escribe un 
bit de información a la localidad de memoria seleccionada. 

La siguiente figura muestra el envío y recepción de la 
información: 
 

   
 
EJEMPLO DE ESCRITURA Y LECTURA DE LOS REGISTROS 
DEL RELOJ/CALENDARIO. 
 
Para este ejemplo se utilizará como base el lenguaje de 
programación de los microcontroladores PIC de la familia 
intermedia.  
 
El siguiente código ejemplifica como enviar los datos 
correspondientes a las horas, los minutos y los segundos 
hacia el DS1302, y posteriormente leer esos datos desde el 
DS1302 y guardarlos en la memoria RAM del 
microcontrolador. 
 
 
 



 
 
 
P I C M I C R O E S T D U D I O 
 
IO EQU 1; Son los pines del puerto donde están conectadas las terminales de comunicación 
RST EQU 2; Estas pueden variar según el circuito y la disposición de pines 
SCLK EQU 3;  
UNDATO EQU 0X20; Localidad de memoria donde se almacenará temporalmente el dato recibido del DS1302 
SEGUN EQU 0X21; Localidad que almacena los segundos, leídos del DS1302 
MIN EQU 0X22; Localidad que almacena los minutos, leídos del DS1302 
HOR EQU 0X23; Localidad que almacena las horas, leídos del DS1302 
 
 ORG 0 
  
 BSF STATUS,RP0; Banco 1 
 CLRF TRISB; Configura el Puerto B como salida 
 BCF STATUS,RP0; Banco 0 
 
 
;//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 
;OPERACIONES DE ESCRITURA DEL DS1302 
 
;************************************************* 
;Deshabilita el Write Protect 
 
 MOVLW 0X8E ;Comando que apunta al registro de control para una escritura 
 CALL COMRTC ;Llama a la rutina que envía el comando hacia el DS1302 
 MOVLW 0X00 ;Este dato deshabilita el Write Protect para permitir futuras escrituras 
 CALL ENVIARTC1;Llama a la rutina que envía el dato hacia el DS1302 
 
;*************************************************** 
;ESCRIBE el numero 0x35 a al registro correspondiente de los segundos del DS1302 
 
 MOVLW  0X80 ; POSICIONA LA LOCALIDAD DE LOS SEGUNDOS PARA ESCRITURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 MOVLW 0X35 ;ENVIA EL DATO 0X35 A LOS SEGUNDOS 
 CALL ENVIARTC1; Llama a la rutina que envía el dato hacia el DS1302  
 
;**************************************************** 
;ESCRIBE el numero 0x22 a al registro correspondiente de los minutos del DS1302 
 MOVLW  0X82 ; POSICIONA LA LOCALIDAD DE LOS MINUTOS PARA ESCRITURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 MOVLW 0X22 ;ENVIA EL DATO 0X22 A LOS MINUTOS 
 CALL ENVIARTC1; Llama a la rutina que envía el dato hacia el DS1302  
 
;**************************************************** 
;ESCRIBE el numero 0x13 al registro correspondiente de las horas del DS1302 
 MOVLW  0X84 ; POSICIONA LA LOCALIDAD DE LAS HORAS PARA ESCRITURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 MOVLW 0X13 ;ENVIA EL DATO 0X13 A LAS HORAS 
 CALL ENVIARTC1; Llama a la rutina que envía el dato hacia el DS1302 
 BCF PORTB,RST ;Deshabilita las comunicaciones  
 
;/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 
;OPERACIONES DE LECTURA 
 
;*************************************************** 
;Lee el registro correspondiente de los segundos del DS1302 
 
 MOVLW  0X81 ; POSICIONA LA LOCALIDAD DE LOS SEGUNDOS PARA LECTURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 CALL RECIBRTC1; Llama a la rutina que recibe el dato enviado desde el DS1302  
 MOVF UNDATO,W; Mueve el dato recibido a W 
 MOVWF SEGUN ; Guarda el dato en la variable SEGUN 
 
;*************************************************** 

;Lee el registro correspondiente de los minutos del DS1302 
 
 MOVLW  0X83 ; POSICIONA LA LOCALIDAD DE LOS MINUTOS PARA LECTURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 CALL RECIBRTC1; Llama a la rutina que recibe el dato enviado desde el DS1302  
 MOVF UNDATO,W; Mueve el dato recibido a W 
 MOVWF MIN ; Guarda el dato en la variable MIN 
;*************************************************** 
;Lee el registro correspondiente de las horas del DS1302 
 
 MOVLW  0X85 ; POSICIONA LA LOCALIDAD DE LAS HORAS PARA LECTURA 
 CALL COMRTC; Llama la rutina que envía el comando 
 CALL RECIBRTC1; Llama a la rutina que recibe el dato enviado desde el DS1302  
 MOVF UNDATO,W; Mueve el dato recibido a W 
 MOVWF HOR ; Guarda el dato en la variable HOR 
 
;DE ESTA MANERA ES COMO SE ESCRIBEN Y SE LEEN LOS REGISTROS DEL DS1302. USTED PUEDE MANIPULAR 
LOS DATOS RECIBIDOS, QUE SE HAN GUARDADO EN LA RAM, PARA MOSTRARLOS EN UN DISPLAY  
 
FIN GOTO FIN 
 
COMRTC  MOVWF ALFA 
 BSF STATUS,RP0 
 BCF TRISB,IO ;CONFIGURA EL PUERTO COMO SALIDA 
 BCF STATUS,RP0 
 BSF PORTB,RST ;HABILITA LA COMUNICACIÓN 
 MOVLW 0X08 
 MOVWF ROTA ;ESTE ES UN CONTADOR 
COMRTC1 BSF PORTB,IO ;PONE A 0 O A 1 EL PIN IO DEL PUERTO 
 BTFSS ALFA,0 ;DEPENDIENDO EL VALOR DEL BIT 0 DEL DATO A ENVIAR 
 BCF PORTB,IO 
 BSF PORTB,SCLK; FLANCO DE SUBIDA DE SCLK 
 BCF PORTB,SCLK; FLANCO DE BAJADA DE SCLK 
 DECFSZ ROTA,F ; 
 GOTO FALTAN 
 RETURN 
FALTAN RRF ALFA,F ;ROTA EL DATO A ENVIAR 
 GOTO COMRTC1 
 
;************************************************* 
;RUTINA QUE ENVIA UN DATO AL DS1302 DESPUES DEL COMANDO 
ENVIARTC1 
 MOVWF ALFA 
 MOVLW 0X08 
 MOVWF ROTA  
 BSF STATUS,RP0 
 BCF TRISB,IO 
 BCF STATUS,RP0   
  
OTRO2 BSF PORTB,IO;PONE A 0 O A 1 EL PIN IO DEL PUERTO 
 BTFSS ALFA,0 ;DEPENDIENDO EL VALOR DEL BIT 0 DEL DATO A ENVIAR 
 BCF PORTB,IO; FLANCO DE SUBIDA DE SCLK 
 BCF PORTB,SCLK; FLANCO DE BAJADA DE SCLK 
 DECFSZ ROTA,F 
 GOTO FALTAN12 
 BCF PORTB,RST 
 RETURN 
FALTAN12 RRF ALFA,F 
 GOTO OTRO2 
 
;******************************************************* 

;RUTINA QUE RECIBE UN DATO DESDE EL DS1302 
 
RECIBRTC1 BSF STATUS,RP0 
 BSF TRISB,IO 
 BCF STATUS,RP0 
 MOVLW 0X08 
 MOVWF ROTA 
 
OTRO22 BSF UNDATO,7 ;Registro donde se guarda bit a bit el dato proveniente del DS1302 
 BTFSS PORTB,IO 
 BCF UNDATO,7 
 BSF PORTB,SCLK 
 BCF PORTB,SCLK; SEÑAL DE RELOJ 
 DECFSZ ROTA,F 
 GOTO FALTAN22 
 BCF PORTB,RST 
 RETURN 
FALTAN22 BCF STATUS,C 
 RRF UNDATO,F 
 GOTO OTRO22 
 
 END 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 


